Lezione 4 (parte prima)

Enrico Bertolazzi

```
> # spline cubiche, derivazione
> # definizione della derivata seconda del k-esimo
  # tratto di spline. M[k] = valore della derivata
  # seconda nel k-esimo nodo.
  Sxx := (x,k) -> (M[k-1]*(X[k]-x)+M[k]*(x-X[k-1]))/(X[k]-X[k-1]) ;
```

$$Sxx := (x,k) \rightarrow \frac{M_{k-1}\left(X_k - x\right) + M_k\left(x - X_{k-1}\right)}{X_k - X_{k-1}} \quad (1)$$

```
> # tramite integrazione costruisco la derivata prima
  # della spline (a meno di una costante)
  Sx := unapply(int(Sxx(x,k),x),(x,k)) ;
```

$$Sx := (x,k) \rightarrow \frac{M_{k-1}\left(X_k x - \frac{1}{2}x^2\right) + M_k\left(\frac{1}{2}x^2 - X_{k-1}x\right)}{X_k - X_{k-1}} \quad (2)$$

```
> # tramite integrazione costruisco la
  # spline (a meno di una retta)
  S := unapply(int(Sx(x,k),x),(x,k));
```

$$S := (x,k) \rightarrow \frac{M_{k-1}\left(\frac{1}{2}X_k x^2 - \frac{1}{6}x^3\right) + M_k\left(\frac{1}{6}x^3 - \frac{1}{2}X_{k-1}x^2\right)}{X_k - X_{k-1}} \quad (3)$$

```
> # aggiungo una retta a S per costruire il
  # generico tratto di spline
  SGEN := unapply( S(x,k) + A + B * x, (x,k) ) ;
```

$$SGEN := (x,k) \rightarrow \frac{M_{k-1}\left(\frac{1}{2}X_k x^2 - \frac{1}{6}x^3\right) + M_k\left(\frac{1}{6}x^3 - \frac{1}{2}X_{k-1}x^2\right)}{X_k - X_{k-1}} + A + Bx \quad (4)$$

```
> # condizioni di interpolazione
  EQK1 := SGEN(X[k-1],k) = Y[k-1] ;
  EQK  := SGEN(X[k],k)   = Y[k] ;
```

$$EQK1 := \frac{M_{k-1}\left(\frac{1}{2}X_k X_{k-1}^2 - \frac{1}{6}X_{k-1}^3\right) - \frac{1}{3}M_k X_{k-1}^3}{X_k - X_{k-1}} + A + B X_{k-1} = Y_{k-1} \quad (5)$$

$$EQK := \frac{\frac{1}{3}M_{k-1}X_k^3 + M_k\left(\frac{1}{6}X_k^3 - \frac{1}{2}X_{k-1}X_k^2\right)}{X_k - X_{k-1}} + A + B X_k = Y_k$$

```
> # ricavo A e B con le condizioni di interpolazione
```

```
res := solve({EQK,EQK1},{A,B});
```

$$res := \left\{A = \frac{1}{6}\,\frac{1}{X_k - X_{k-1}}\left(M_k X_{k-1} X_k^2 + 2 X_k^2 M_{k-1} X_{k-1} - 2 X_k M_k X_{k-1}^2 + 6 Y_{k-1} X_k - M_{k-1} X_k X_{k-1}^2 - 6 Y_k X_{k-1}\right), B = -\frac{1}{6}\,\frac{1}{X_k - X_{k-1}}\left(X_k^2 M_k + 2 X_k^2 M_{k-1} - 2 X_k M_k X_{k-1} - 6 Y_k + 2 X_k M_{k-1} X_{k-1} - 2 M_k X_{k-1}^2 + 6 Y_{k-1} - M_{k-1} X_{k-1}^2\right)\right\} \tag{6}$$

```
> # sostituisco A e B e costruisco il tratto di cubica generico
  SGENB := unapply(collect(expand(subs(res,SGEN(x,k))),x),(x,k)) ;
```

$$SGENB := (x,k) \to \left(-\frac{1}{6}\,\frac{M_{k-1}}{X_k - X_{k-1}} + \frac{\frac{1}{6} M_k}{X_k - X_{k-1}}\right) x^3 + \left(-\frac{1}{2}\,\frac{M_k X_{k-1}}{X_k - X_{k-1}} + \frac{\frac{1}{2} M_{k-1} X_k}{X_k - X_{k-1}}\right) x^2 + \left(\frac{1}{3}\,\frac{M_k X_{k-1}^2}{X_k - X_{k-1}} - \frac{Y_{k-1}}{X_k - X_{k-1}} + \frac{Y_k}{X_k - X_{k-1}} - \frac{1}{3}\,\frac{X_k M_{k-1} X_{k-1}}{X_k - X_{k-1}} - \frac{1}{6}\,\frac{X_k^2 M_k}{X_k - X_{k-1}} - \frac{1}{3}\,\frac{X_k^2 M_{k-1}}{X_k - X_{k-1}} + \frac{\frac{1}{3} X_k M_k X_{k-1}}{X_k - X_{k-1}} + \frac{\frac{1}{6} M_{k-1} X_{k-1}^2}{X_k - X_{k-1}}\right) x + \frac{\frac{1}{6} M_k X_{k-1} X_k^2}{X_k - X_{k-1}} + \frac{\frac{1}{3} X_k^2 M_{k-1} X_{k-1}}{X_k - X_{k-1}} - \frac{1}{3}\,\frac{X_k M_k X_{k-1}^2}{X_k - X_{k-1}} + \frac{Y_{k-1} X_k}{X_k - X_{k-1}} - \frac{1}{6}\,\frac{M_{k-1} X_k X_{k-1}^2}{X_k - X_{k-1}} - \frac{Y_k X_{k-1}}{X_k - X_{k-1}} \tag{7}$$

```
> # semplifico SGENB tramite le sostituzioni
  # h[k]=X[k]-X[k-1] e z=x-X[k-1]
  SGENC:= unapply(collect(expand(subs(X[k]=X[k-1]+h[k],x=X[k-1]+z,
  SGENB(x,k))),z),(z,k)) ;
```

$$SGENC := (z,k) \to \left(-\frac{1}{6}\,\frac{M_{k-1}}{h_k} + \frac{\frac{1}{6} M_k}{h_k}\right) z^3 + \frac{1}{2} M_{k-1} z^2 + \left(\frac{Y_k}{h_k} - \frac{Y_{k-1}}{h_k} - \frac{1}{6} h_k M_k - \frac{1}{3} h_k M_{k-1}\right) z + Y_{k-1} \tag{8}$$

```
> # costruisco la derivata del tratto di cubica generico
  SGENz := unapply(diff(SGENC(x-X[k-1],k),x),(x,k)) ;
```

(9)

$$SGENz := (x,k) \to 3\left(-\frac{1}{6}\,\frac{M_{k-1}}{h_k} + \frac{\frac{1}{6}\,M_k}{h_k}\right)\left(x - X_{k-1}\right)^2 + M_{k-1}\left(x - X_{k-1}\right) + \frac{Y_k}{h_k} - \frac{Y_{k-1}}{h_k} - \frac{1}{6}\,h_k\,M_k - \frac{1}{3}\,h_k\,M_{k-1} \tag{9}$$

```
> # condizione di continuita della derivata prima nel punto X[k]
  EQT := collect(SGENz(X[k],k) - SGENz(X[k],k+1)=0,{M[k-1],M[k],M
  [k+1]}) ;
```

$$EQT := \left(X_k - X_{k-1} - \frac{1}{2}\,\frac{\left(X_k - X_{k-1}\right)^2}{h_k} - \frac{1}{3}\,h_k\right)M_{k-1} + \left(\frac{1}{2}\,\frac{\left(X_k - X_{k-1}\right)^2}{h_k} - \frac{1}{6}\,h_k + \frac{1}{3}\,h_{k+1}\right)M_k - \frac{Y_{k+1}}{h_{k+1}} + \frac{Y_k}{h_k} - \frac{Y_{k-1}}{h_k} + \frac{Y_k}{h_{k+1}} + \frac{1}{6}\,h_{k+1}\,M_{k+1} = 0 \tag{10}$$

```
> # semplifico la soluzione
  subs(X[k]=X[k-1]+h[k],EQT);
```

$$\frac{1}{6}\,h_k\,M_{k-1} + \left(\frac{1}{3}\,h_k + \frac{1}{3}\,h_{k+1}\right)M_k - \frac{Y_{k+1}}{h_{k+1}} + \frac{Y_k}{h_k} - \frac{Y_{k-1}}{h_k} + \frac{Y_k}{h_{k+1}} + \frac{1}{6}\,h_{k+1}\,M_{k+1} = 0 \tag{11}$$

```
> # questa ultima equazione e' una riga del sistema tridiagonale
  # una volta risolto e sostituendo i valori degli M nei tratti di
  # cubica ottengo la spline cercata.
```